

```

// these are our calibration values stored in flash
dim xminf as flash, xmaxf as flash, yminf as flash, ymaxf as flash
//
// these are our shadow calibration values stored in ram for speed
dim xminr, xmaxr, yminr, ymaxr
//
// this is when we last updated flash
dim secondsf
secondsf = seconds
//
// this is our interrupt pin indicating measurement complete
dim int as pin an3 for digital input
//
// this is our magnetometer i2c address
dim addr
addr = 0xe
//
// this is the calibrated magnetometer heading
dim xa, ya, deg
//
// initialize the 3110 and start taking interrupts
gosub mag3110_init addr
on int do gosub loop
halt
//
// poll the magnetometer x and y values and print the heading
sub loop
gosub mag3110_poll addr, xa, ya, deg
print "x =", xa, "y =", ya, "deg =", deg
endsub
//
// get the magnetometer calibration values
sub mag3110_getcal xmin, xmax, ymin, ymax
// if our shadow values are not set...
if xminr==0&&xmaxr==0&&yminr==0&&ymaxr==0 then
// read flash
// N.B. we could just as easily read eeprom or whatever
xmin = xminf, xmax = xmaxf, ymin = yminf, ymax = ymaxf
else
// read shadow
xmin = xminr, xmax = xmaxr, ymin = yminr, ymax = ymaxr
endif
endsub
//
// set the magnetometer calibration values
sub mag3110_setcal xmin, xmax, ymin, ymax
// update shadow
xminr = xmin, xmaxr = xmax, yminr = ymin, ymaxr = ymax
// if it has been ten seconds since we last updated flash...
if seconds>secondsf+10 then
// update flash
// N.B. we could just as easily update eeprom or whatever
xminf = xmin, xmaxf = xmax, yminf = ymin, ymaxf = ymax
secondsf = seconds
print "update"
endif
endsub

```